<commit_before>use v6;

=begin pod

=TITLE 99 bottles of beer

=AUTHOR <NAME>

Print the lyrics to the
L<99 bottles of beer|http://en.wikipedia.org/wiki/99_Bottles_of_Beer>
song.

=end pod

my @bottles = ((99...2) X~ ' bottles'),
              '1 bottle',
              'no more bottles',
              '99 bottles';

my @actions = 'Take one down and pass it around' xx 99,
              'Go to the store and buy some more';

for @bottles Z @actions Z @bottles[1..*] {
    say "$^a of beer on the wall, $^a of beer.
$^b, $^c of beer on the wall.\n".tc;
}

# vim: expandtab shiftwidth=4 ft=perl6<commit_msg>Fix 99 bottles of beer.  Dont know if best solution to GLR but at least it works.<commit_after>use v6;

=begin pod

=TITLE 99 bottles of beer

=AUTHOR <NAME>

Print the lyrics to the
L<99 bottles of beer|http://en.wikipedia.org/wiki/99_Bottles_of_Beer>
song.

=end pod

my @bottles = (flat ((99...2) X~ ' bottles'),
              '1 bottle',
              'no more bottles',
              '99 bottles');

my @actions = (flat 'Take one down and pass it around' xx 99,
              'Go to the store and buy some more');

for flat @bottles Z @actions Z @bottles[1..*] {
    say "$^a of beer on the wall, $^a of beer.
$^b, $^c of beer on the wall.\n".tc;
}

# vim: expandtab shiftwidth=4 ft=perl6
{'fraction_non_alphanumeric': 0.1308698999230177, 'fraction_numerical': 0.029253271747498075, 'mean_word_length': 2.5616438356164384, 'pattern_counts': {'":': 0, '<': 7, '<?xml version=': 0, '>': 7, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2458963', 'n_tokens_mistral': 492, 'n_tokens_neox': 459, 'n_words': 180}